

Formalization of the Fundamental Group in Untyped Set Theory Using Auto2

Bohua Zhan^(✉)

Massachusetts Institute of Technology, Cambridge, USA
bzhan@mit.edu

Abstract. We present a new framework for formalizing mathematics in untyped set theory using auto2. Using this framework, we formalize in Isabelle/FOL the entire chain of development from the axioms of set theory to the definition of the fundamental group for an arbitrary topological space. The auto2 prover is used as the sole automation tool, and enables succinct proof scripts throughout the project.

1 Introduction

Auto2, introduced by the author in [17], is a proof automation tool for the proof assistant Isabelle. It is designed to be a powerful, extensible prover that can consistently solve “routine” tasks encountered during a proof, thereby enabling a style of formalization using succinct proof scripts written in a custom, purely declarative language.

In this paper, we present an application of auto2 to formalization of mathematics in untyped set theory¹. In particular, we discuss the formalization in Isabelle/FOL of the entire chain of development from the axioms of set theory to the definition of the fundamental group for an arbitrary topological space. Along the way, we discuss several improvements to auto2 as well as strategies of usage that allow us to work effectively with untyped set theory.

The contribution of this paper is two-fold. First, we demonstrate that the auto2 system is capable of independently supporting proof developments on a relatively large scale. In the previous paper, several case studies for auto2 were given in Isabelle/HOL. Each case study is at most several hundred lines long, and the use of auto2 is mixed with the use of other Isabelle tactics, as well as proof scripts provided by Sledgehammer. In contrast, the example we present in this paper is a unified development consisting of over 13,000 lines of theory files and 3,500 lines of ML code (not including the core auto2 program). The auto2 prover is used exclusively starting from basic set theory.

Second, we demonstrate one way to manage the additional complexity in proofs that arise when working with untyped set theory. For a number of reasons, untyped set theory is considered to be difficult to work with. For example, everything is represented as sets, including objects such as natural numbers that

¹ Code available at <https://github.com/bzhan/auto2>.

we usually do not think of as sets. Moreover, statements of theorems tend to be longer in untyped set theory than in typed theories, since assumptions that would otherwise be included in type constraints must now be stated explicitly. In this paper, we show that with appropriate definitions of basic concepts and setup for automation, all these complexities can be managed, without sacrificing the inherent flexibility of the logic.

We now give an outline for the rest of the paper. In Sect. 2, we sketch our choice of definitions of basic concepts in axiomatic set theory. In particular, we describe how to use tuples to realize extensible records, and build up the hierarchy of algebraic structures. In Sect. 3, we review the main ideas of the auto2 system, and describe several additional features, as well as strategies of usage, that allow us to manage the additional complexities of untyped set theory.

In Sect. 4, we give two examples of proof scripts using auto2, taken from the proofs of the Schroeder-Bernstein theorem and a challenge problem in analysis from Lasse Rempe-Gillen. In Sect. 5, we describe our main example, the definition of the fundamental group, in detail. Given a topological space X and a base point x on X , the fundamental group $\pi_1(X, x)$ is defined on the quotient of the set of loops in X based at x , under the equivalence relation given by path homotopy. Multiplication on $\pi_1(X, x)$ comes from joining two loops end-to-end. Formalizing this definition requires reasoning about algebraic and topological structures, equivalence relations, as well as continuous functions on real numbers. We believe this is a sufficiently challenging task with which to test the maturity of our framework, although it has been achieved before in the Mizar system. HOL Light and Isabelle/HOL also formalized the essential ideas on path homotopy. We review these and other related works in Sect. 6, and conclude in Sect. 7.

2 Basic Constructions in Set Theory

We now discuss our choice of definitions of basic concepts, starting with the choice of logic. Our development is based on the FOL (first-order logic) instantiation of Isabelle. The initial parts are similar to those in Isabelle/ZF, and we refer to [13, 14] for detailed explanations.

The only Isabelle types available are i for sets, o for propositions (booleans), and function types formed from them. We call objects with types other than i and o *meta-functions*, to distinguish them from functions defined within set theory (which have type i). It is possible to define higher-order meta-functions in FOL, and supply them with arguments in the form of lambda expressions. Theorems can be quantified over variables with functional type at the outermost level. These can be thought of as theorem-schemas in a first-order theory. However, one can only quantify over variables of type i inside the statement of a theorem, and the only equalities defined within FOL are those between types i (notation $\cdot = \cdot$) and o (notation $\cdot \longleftrightarrow \cdot$). In practice, these restrictions mean that any functions that we wish to consider as first-class objects must be defined as set-theoretic functions.

2.1 Axioms of Set Theory

For uniformity of presentation, we start our development from FOL rather than theories in Isabelle/ZF. However, the list of axioms we use is mostly the same. The only main addition is the axiom of global choice, which we use as an easier-to-apply version of the axiom of choice. Note that as in Isabelle/ZF, several of the axioms introduce new sets or meta-functions, and declare properties satisfied by them. The exact list of axioms is as follows:

- extension*: $"\forall z. z \in x \longleftrightarrow z \in y \implies x = y"$
- empty_set*: $"x \notin \emptyset"$
- collect*: $"x \in \text{Collect}(A,P) \longleftrightarrow (x \in A \wedge P(x))"$
- upair*: $"x \in \text{Upair}(y,z) \longleftrightarrow (x = y \vee x = z)"$
- union*: $"x \in \bigcup C \longleftrightarrow (\exists A \in C. x \in A)"$
- power*: $"x \in \text{Pow}(S) \longleftrightarrow x \subseteq S"$
- replacement*: $"\forall x \in A. \forall y z. P(x,y) \wedge P(x,z) \longrightarrow y = z \implies b \in \text{Replace}(A,P) \longleftrightarrow (\exists x \in A. P(x,b))"$
- foundation*: $"x \neq \emptyset \implies \exists y \in x. y \cap x = \emptyset"$
- infinity*: $"\emptyset \in \text{Inf} \wedge (\forall y \in \text{Inf}. \text{succ}(y) \in \text{Inf})"$
- choice*: $"\exists x. x \in S \implies \text{Choice}(S) \in S"$

Next, we define several basic constructions in set theory. They are summarized in the following table. See [13] for more explanations.

Notation	Definition
<i>THE</i> $x. P(x)$	$\bigcup (\text{Replace}(\{\emptyset\}, \lambda x y. P(y)))$
$\{b(x). x \in A\}$	$\text{Replace}(A, \lambda x y. y = b(x))$
<i>SOME</i> $x \in A. P(x)$	$\text{Choice}(\{x \in A. P(x)\})$
$\langle a, b \rangle$	$\{\{a\}, \{a, b\}\}$
<i>fst</i> (p)	<i>THE</i> $a. \exists b. p = \langle a, b \rangle$
<i>snd</i> (p)	<i>THE</i> $b. \exists a. p = \langle a, b \rangle$
$\langle a_1, \dots, a_n \rangle$	$\langle a_1, \langle a_2, \langle \dots, a_n \rangle \rangle \rangle$
<i>if</i> P <i>then</i> a <i>else</i> b	<i>THE</i> $z. P \wedge z=a \vee \neg P \wedge z=b$
$\bigcup_{a \in I}. X$	$\bigcup \{X(a). a \in I\}$
$A \times B$	$\bigcup_{x \in A}. \bigcup_{y \in B}. \{\langle x, y \rangle\}$

2.2 Extensible Records as Tuples

We now consider the problem of representing records. In our framework, records are used to represent functions, algebraic and topological structures, as well as morphisms between structures. It is often advantageous for records of different types to share certain fields. For example, groups and rings should share the multiplication operator, rings and ordered rings should share both addition and multiplication operators, and so on.

It is well-known that when formalizing mathematics using set theory, records can be represented as tuples. To achieve sharing of fields, the key idea is to assign each shared field a fixed position in the tuple.

We begin with the example of functions. A function is a record consisting of a source set (domain), a target set (codomain), and the graph of the function. In

particular, we consider two functions with the same graph but different target sets to be different functions (another structure called *family* is used to represent functions without specified target set). The three fields are assigned to the first three positions in the tuple:

definition `"source(F) = fst(F)"`
definition `"target(F) = fst(snd(F))"`
definition `"graph(F) = fst(snd(snd(F)))"`

A function with source S , target T , and graph G is represented by the tuple $\langle S, T, G, \emptyset \rangle$ (we append an \emptyset at the end so the definition of *graph* works properly). For G to actually represent a function, it must satisfy the conditions for a functional graph:

definition `func_graphs :: "i \Rightarrow i \Rightarrow i" where`
`"func_graphs(X, Y) = {G \in Pow(X \times Y). ($\forall a \in X. \exists ! y. \langle a, y \rangle \in G$)}"`

The set of all functions from S to T (denoted $S \rightarrow T$) is then given by:

definition `function_space :: "i \Rightarrow i \Rightarrow i" (infixr " \rightarrow " 60) where`
`"A \rightarrow B = { $\langle A, B, G, \emptyset \rangle. G \in func_graphs(A, B)$ }"`

Functions can be created using the following constructor. Note this is a higher-order meta-function. The argument b can be supplied by a lambda expression.

definition `Fun :: "[i, i, i \Rightarrow i] \Rightarrow i" where`
`"Fun(A, B, b) = $\langle A, B, \{p \in A \times B. snd(p) = b(fst(p))\}, \emptyset \rangle"$`

Evaluation of a function f at x (denoted $f \backslash x$) is then defined as:

definition `feval :: "i \Rightarrow i \Rightarrow i" (infixl " \backslash " 90) where`
`"f \backslash x = (THE y. $\langle x, y \rangle \in graph(f)$)"`

2.3 Algebraic Structures

The second major use of records is to represent algebraic structures. In our framework, we will define structures such as groups, abelian groups, rings, and ordered rings. The carrier set of a structure is assigned to the first position. The order relation, additive data, and multiplicative data are assigned to the third, fourth, and fifth position, respectively. This is expressed as follows:

definition `"carrier(S) = fst(S)"`
definition `"order_graph(S) = fst(snd(snd(S)))"`
definition `"zero(S) = fst(fst(snd(snd(snd(S)))))"`
definition `"plus_fun(S) = snd(fst(snd(snd(snd(S)))))"`
definition `"one(S) = fst(fst(snd(snd(snd(snd(S))))))"`
definition `"times_fun(S) = snd(fst(snd(snd(snd(snd(S))))))"`

Here *order_graph* is a subset of $S \times S$, and *plus_fun*, *times_fun* are elements of $S \times S \rightarrow S$. Hence, the operators \leq , $+$, and $*$ can be defined as follows:

definition $"1e(R, x, y) \longleftrightarrow \langle x, y \rangle \in \text{order_graph}(R)"$

definition $"plus(R, x, y) = \text{plus_fun}(R) \setminus \langle x, y \rangle"$

definition $"times(R, x, y) = \text{times_fun}(R) \setminus \langle x, y \rangle"$

These are abbreviated to $x \leq_R y$, $x +_R y$, and $x *_R y$, respectively (in both theory files and throughout this paper, we use $*$ to denote multiplication in groups and rings, and \times to denote product on sets and other structures). We also abbreviate $x \in \text{carrier}(S)$ to $x \in S$.

The constructor for group-like structures is as follows:

definition $\text{Group} :: "[i, i, i \Rightarrow i \Rightarrow i] \Rightarrow i"$ where

$"\text{Group}(S, u, f) = \langle S, \emptyset, \emptyset, \emptyset, \langle u, \lambda p \in S \times S. f(\text{fst}(p), \text{snd}(p)) \rangle \in S, \emptyset \rangle"$

The following predicate asserts that a structure contains *at least* the fields of a group-like structure, with the right membership properties ($\mathbf{1}_G$ abbreviates $\text{one}(G)$):

definition $\text{is_group_raw} :: "i \Rightarrow o"$ where

$"\text{is_group_raw}(G) \longleftrightarrow$

$\mathbf{1}_G \in G \wedge \text{times_fun}(G) \in \text{carrier}(G) \times \text{carrier}(G) \rightarrow \text{carrier}(G)$

To check whether such a structure is in fact a monoid/group, we use the following predicates:

definition $\text{is_monoid} :: "i \Rightarrow o"$ where

$"\text{is_monoid}(G) \longleftrightarrow \text{is_group_raw}(G) \wedge$

$(\forall x \in G. \forall y \in G. \forall z \in G. (x *_G y) *_G z = x *_G (y *_G z)) \wedge$

$(\forall x \in G. \mathbf{1}_G *_G x = x \wedge x *_G \mathbf{1}_G = x)"$

definition $\text{units} :: "i \Rightarrow i"$ where

$"\text{units}(G) = \{x \in G. (\exists y \in G. y *_G x = \mathbf{1}_G \wedge x *_G y = \mathbf{1}_G)\}"$

definition $\text{is_group} :: "i \Rightarrow o"$ where

$"\text{is_group}(G) \longleftrightarrow \text{is_monoid}(G) \wedge \text{carrier}(G) = \text{units}(G)"$

Note these definitions are meaningful on any structure that has multiplicative data. Likewise, we can define a predicate is_abgroup for abelian groups, that is meaningful for any structure that has additive data. These can be combined with distributive properties to define the predicate for a ring:

definition $\text{is_ring} :: "i \Rightarrow o"$ where

$"\text{is_ring}(R) \longleftrightarrow (\text{is_ring_raw}(R) \wedge \text{is_abgroup}(R) \wedge \text{is_monoid}(R) \wedge$

$\text{is_left_distrib}(R) \wedge \text{is_right_distrib}(R) \wedge \mathbf{0}_R \neq \mathbf{1}_R)"$

Likewise, we can define the predicate for ordered rings, and constructors for such structures. Structures are used to represent the hierarchy of numbers: we let *nat int*, *ra*, and *real* denote the *set* of natural numbers, integers, etc., while \mathbb{N} , \mathbb{Z} , \mathbb{Q} , and \mathbb{R} denote the corresponding structures. Hence, addition on natural numbers is denoted by $x +_{\mathbb{N}} y$, addition on real numbers by $x +_{\mathbb{R}} y$, etc. We can also state and prove theorems such as $\text{is_ord_field}(\mathbb{R})$, which contains all proof obligations for showing that the real numbers form an ordered field.

2.4 Morphism Between Structures

Finally, we discuss morphisms between structures. Morphisms can be considered as an *extension* of functions, with additional fields specifying structures on the source and target sets. The two additional fields are assigned to the fourth and fifth positions in the tuple:

definition `"source_str(F) = fst(snd(snd(snd(F))))"`

definition `"target_str(F) = fst(snd(snd(snd(snd(F)))))"`

The constructor for a morphism is as follows (here S and T are the source and target structures, while the source and target sets are automatically derived):

definition `Mor :: "[i, i, i ⇒ i] ⇒ i" where`
`"Mor(S,T,b) = (let A = carrier(S) in let B = carrier(T) in`
`{A, B, {p∈A×B. snd(p) = b(fst(p))}, S, T, ∅)"`

The space of morphisms (denoted $S \rightarrow T$) is given by:

definition `mor_space :: "i ⇒ i ⇒ i" (infix "→" 60) where`
`"mor_space(S,T) = (let A = carrier(S) in let B = carrier(T) in`
`{⟨A,B,G,S,T,∅⟩. G∈func_graphs(A,B)}"`

Note the notation $f \backslash x$ for evaluation still works for morphisms. Several other concepts defined in terms of evaluation, such as image and inverse image, continue to be valid for morphisms as well, as are lemmas about these concepts. However, operations that construct new morphisms, such as inverse and composition, must be redefined. We will use $g \circ f$ to denote the composition of two functions, and $g \circ_m f$ to denote the composition of two morphisms.

Having morphisms store the source and target structures means we can define properties such as homomorphism on groups as a predicate:

definition `is_group_hom :: "i ⇒ o" where`
`"is_group_hom(f) ⇔ (let S = source_str(f) in let T = target_str(f) in`

`is_morphism(f) ∧ is_group(S) ∧ is_group(T) ∧`
`(∀ x∈.S. ∀ y∈.S. f \ (x *_S y) = f \ x *_T f \ y))"`

The following lemma then states that the composition of two homomorphisms is a homomorphism (this is proved automatically using `auto2`):

lemma `group_hom_compose:`
`"is_group_hom(f) ⇒ is_group_hom(g) ⇒`
`target_str(f) = source_str(g) ⇒ is_group_hom(g ◦_m f)"`

3 Auto2 in Untyped Set Theory

In this section, we describe several additional features of `auto2`, as well as general strategies of using it to manage the complexities of untyped set theory.

We begin with an overview of the `auto2` system (see [17] for details). `Auto2` is a theorem prover packaged as a tactic in Isabelle. It works with a collection of rules of reasoning called *proof steps*. New proof steps can be added at any time within an Isabelle theory. They can also be deleted at any time, although it is rarely necessary to add and delete the same proof step more than once. In general, when building an Isabelle theory, the user is responsible for specifying, by adding proof steps, how to use the results proved in that theory. In return, the user no longer needs to worry about invoking these results by name in future developments.

The overall algorithm of `auto2` is as follows. First, the statement to be proved is converted into contradiction form, so the task is always to derive a contradiction from a list of assumptions. During the proof, `auto2` maintains a list of *items*, the two most common types of which are propositions (that are derived from the assumptions) and terms (that have appeared so far in the proof). Each item resides in a *box*, which can be thought of as a subcase of the statement to be proved (the box corresponding to the original statement is called the *home box*). A proof step is a function that takes as input one or two items, and outputs either new items, new cases, or the action of shadowing one of the input items, or resolving a box by proving a contradiction in that box.

The main loop of the algorithm repeatedly applies the current collection of proof steps and adds any new items and cases in a best-first-search manner, until some proof step derives a contradiction in the home box. In addition to the list of items, `auto2` also maintains several tables. The most important of which is the *rewrite table*, which keeps track of the list of currently known equalities (not containing arbitrary variables), and maintains the congruence closure of these equalities. There are two other tables: the property table and the well-form table, which we will discuss later in this section.

There are two broad categories of proof steps, which we call the *standard* and *special* proof steps in this paper. A standard proof step applies an existing theorem in a specific direction. It matches the input items to one or two patterns in the statement of the theorem, and applies the theorem to derive a new proposition. Here the matching is up to rewriting (*E-matching*) using the rewrite table. A special proof step can have more complex behavior, and is usually written as an ML function. The vast majority of proof steps in our example are standard, although special proof steps also play an important role.

The `auto2` prover is not intended to be complete. For example, it may intentionally apply a theorem in only one of several possible directions, in order to narrow the search space. For more difficult theorems, `auto2` provides a custom language of proof scripts, allowing the user to specify intermediate steps of the proof. Generally, when proving a result using `auto2`, the user will first try to prove it without any scripts, and in case of failure, successively add intermediate steps, perhaps by referring to an informal proof of the result. In case of failure, `auto2` will indicate the first intermediate step that it is unable to prove, as well as what it is able to derive in the course of proving that step. We will show examples of proof scripts in Sect. 4.

The current version of `auto2` can be set up to work with different logics in Isabelle. It contains a core program, for reasoning about predicate logic and equality, that is parametrized over the list of constants and theorems for the target logic. In particular, `auto2` is now set up and tested to work with both HOL and FOL in Isabelle.

3.1 Encapsulation of Definitions

One commonly cited problem with untyped set theory is that every object is a set, including those that are not usually considered as sets. Common examples of the latter include ordered pairs, natural numbers, functions, etc. In informal treatments of mathematics, these definitions are only used to establish some basic properties of the objects concerned. Once these properties are proved, the definitions are never used again.

In formal developments, when automation is used to produce large parts of the proof, one potential problem is that the automation may needlessly expand the original definitions of objects, rather than focusing on their basic properties. This increases the search space and obscures the essential ideas of the proof. Using the ability to delete proof steps in `auto2`, this problem can be avoided entirely. For any definition that we wish to drop in the end, we use the following three-step procedure:

1. The definition is stated and added to `auto2` as rewrite rules.
2. Basic properties of the object being defined are stated and proved. These properties are added to `auto2` as appropriate proof steps.
3. The rewrite rules for the original definition are deleted.

For example, after the definitions concerning the representation of functions as tuples in Sect. 2.2, we prove the following lemmas, and add them as appropriate proof steps (as indicated by the attributes in brackets):

```
lemma lambda_is_function [backward]:
  "∀x∈A. f(x)∈B ⇒ Fun(A,B,f) ∈ A → B"
```

```
lemma beta [rewrite]:
  "F = Fun(A,B,f) ⇒ x ∈ source(F) ⇒ is_function(F) ⇒ F`x = f(x)"
```

```
lemma feval_in_range [typing]:
  "is_function(f) ⇒ x ∈ source(f) ⇒ f`x ∈ target(f)"
```

After proving these (and a few more) lemmas, the rewriting rules for the definitions of `Fun`, `function_space`, `feval`, etc., are removed. Note that all lemmas above are independent of the representation of functions as tuples. Hence, this representation is effectively hidden from the point of view of the prover. Some of the original definitions may be temporarily re-added in rare instances (for example when defining the concept of morphisms).

3.2 Property and Well-Form Tables

In this section, we discuss two additional tables maintained by auto2 during a proof. The property table is already present in the version introduced in [17], but not discussed in that paper. The well-form table is new.

The main motivation for both tables is that for many theorems, especially those stated in an untyped logic, some of its assumptions can be considered as “side conditions”. To give a basic example, consider the following lemma:

lemma *unit_l_cancel*:

```
"is_monoid(G) ==> y ∈. G ==> z ∈. G ==> x *_c y = x *_c z ==>
x ∈ units(G) ==> y = z"
```

In this lemma, the last two assumptions are the “main” assumptions, while the first three are side conditions asserting that the variables in the main assumptions are well-behaved in some sense. In Isabelle/HOL, these side conditions may be folded into type or type-class constraints.

We consider two kinds of side conditions. The first kind, like the first assumption above, checks that one of the variables in the main assumptions satisfy a certain predicate. In Isabelle/HOL, these may correspond to type-class constraints. In auto2, we call these *property assumptions*. More precisely, given any predicate (in FOL this means constant of type $i \Rightarrow o$), we can register it as a property. The *property table* records the list of properties satisfied by each term that has appeared so far in the proof. Properties propagate through equalities: if $P(a)$ is in the property table, and $a = b$ is known from the rewrite table, then $P(b)$ is automatically added to the property table. The user can also add theorems of certain forms as further propagation rules for the property table (we omit the details here).

The second kind of side conditions assert that certain terms occurring in the main assumptions are *well-formed*. We use the terminology of well-formedness to capture a familiar feature of mathematical language: that an expression may make implicit assumptions about its subterms. These conditions can be in the form of type constraints. For example, the expression $a +_R b$ implicitly assumes that a and b are elements in the carrier set of R . However, this concept is much more general. Some examples of well-formedness conditions are summarized in the following table:

Term	Conditions
$\bigcap A$	$A \neq \emptyset$
$f \setminus x$	$x \in \text{source}(f)$
$g \circ f$	$\text{target}(f) = \text{source}(g)$
$g \circ_m f$	$\text{target_str}(f) = \text{source_str}(g)$
$a +_R b$	$a \in R, b \in R$
$\text{inv}(R, a)$	$a \in \text{units}(R)$
$a /_R b$	$a \in R, b \in \text{units}(R)$
$\text{subgroup}(G, H)$	$\text{is_subgroup_set}(G, H)$
$\text{quotient_group}(G, H)$	$\text{is_normal_subgroup_set}(G, H)$

In general, given any meta-function f , any propositional expression in terms of the arguments of f can be registered as a well-formedness condition of f . In particular, well-formedness conditions are not necessarily properties. For example, the condition $a \in . R$ for $a +_R b$ involves two variables and hence is not a property. The *well-form table* records, for every term encountered so far in the proof, the list of its well-formedness conditions that are satisfied. Whenever a new fact is added, `auto2` checks against every known term to see whether it verifies a well-formedness condition of that term.

The property and well-form tables are used in similar ways in standard proof steps. After the proof step matches one or two patterns in the “main” assumptions or conclusion of the theorem that it applies, it checks for the side conditions in the two tables, and proceed to apply the theorem only if all side conditions are found. Of course, this requires proof steps to be re-applied if new properties or well-formedness conditions of a term becomes known.

3.3 Well-Formed Conversions

Algebraic simplification is an important part of any automatic prover. For every kind of algebraic structure, e.g. monoids, groups, abelian groups, and rings, there is a concept of normal form of an expression, and two terms can be equated if they have the same normal form. In untyped set theory, such computation of normal forms is complicated by the fact that the relevant rewriting rules have extra assumptions. For example, the rule for associativity of addition is:

$$\text{is_abgroup}(R) \implies x \in . R \implies y \in . R \implies z \in . R \implies \\ x +_R (y +_R z) = (x +_R y) +_R z$$

The first assumption can be verified at the beginning of the normalization process. The remaining assumptions, however, are more cumbersome. In particular, they may require membership status of terms that arise only during the normalization. For example, when normalizing the term $a +_R (b +_R (c +_R d))$, we may first rewrite it to $a +_R ((b +_R c) +_R d)$. The next step, however, requires $b +_R c \in . R$, where $b +_R c$ does not occur initially and may not have occurred so far in the proof. In typed theories, this poses no problem, since $b + c$ will be automatically given the same type as b and c when the term is created.

In untyped set theory, such membership information must be kept track of and derived when necessary. The concept of well-formed terms provides a natural framework for doing this. Before performing algebraic normalization on a term, we first check for all relevant well-formedness conditions. If all conditions are present, we produce a data structure (of type *wfterm* in Isabelle/ML) containing the certified term as well as theorems asserting well-formedness conditions. A theorem is called a *well-formed rewrite rule* if its main conclusion is an equality, each of its assumptions is a well-formedness condition for terms on the left side of the equality, and it has additional conclusions that verify all well-formedness conditions for terms on the right side of the equality that are not already present in the assumptions. For example, the associativity rule stated above is not yet a

well-formed rewrite rule: there is no justification for $x +_R y \in . R$, which is a well-formedness condition for the term $(x +_R y) +_R z$ on the right side of the equality. The full well-formed rewrite rule is:

$$\begin{aligned} \text{is_abgroup}(R) \implies x \in . R \implies y \in . R \implies z \in . R \implies \\ x +_R (y +_R z) = (x +_R y) +_R z \wedge x +_R y \in . R \end{aligned}$$

Given a well-formed rewrite rule, we can produce a *well-formed conversion* that acts on *wfterm* objects, in a way similar to how equalities produce regular conversions that act on *cterm* objects in Isabelle/ML. Like regular conversions, well-formed conversions can be composed in various ways, and full normalization procedures can be written using the language of well-formed conversions. These normalization procedures in turn form the basis of several special proof steps. We give two examples:

- Given two terms s and t that are non-atomic with respect to operations in R , where R is a monoid (group/abelian group/ring), normalize s and t using the rules for R . If the normalizations are equal, output $s = t$.
- Given two propositions $a \leq_R b$ and $\neg(c \leq_R d)$, where R is an ordered ring. Compare the normalizations of $b -_R a$ and $d -_R c$. If they are equal, output a contradiction.

These proof steps, when combined with proof scripts provided by the user, allow algebraic manipulations to be performed rapidly. They replace the handling of associative-commutative functions for HOL discussed in [17].

3.4 Discussion

We conclude this section with a discussion of our overall approach to untyped set theory, and compare it with other approaches. One feature of our approach is that we do not seek to re-institute a concept of types in our framework, but simply replace type constraints with set membership conditions (or predicates, for constraints that cannot be described by a set). The aim is to fully preserve the flexibility of set-membership as compared to types. Empirically, most of the extra assumptions that arise in the statement of theorems can be taken care of by classifying them as properties or well-formedness conditions. Our approach can be contrasted with that taken by Mizar, which defines a concept of soft types [16] within the core of the system.

Every framework for formalizing modern mathematics need a way to deal with structures. In Mizar, structures are defined in the core of the system as partial functions on selectors [9, 15]. In both Isabelle/HOL and IsarMathLib’s treatment of abstract algebra, structures are realized with extensive use of locales. For Coq, one notable approach is the use of Canonical Structures [10] in the formalization of the Odd Order Theorem. We chose a relatively simple scheme of realizing structures as tuples, which is sufficient for the present purposes. Representing them as partial functions on selectors, as in Mizar, is more complicated but may be beneficial in the long run.

Finally, we emphasize that we do not make any modification to Isabelle/FOL in our development. The concept of well-formed terms, for example, is meaningful only to the automation. The whole of auto2's design, including the ability for users to add new proof steps, follows the LCF architecture. To have confidence in the proofs, one only need to trust the existing Isabelle system, the ten axioms stated in Sect. 2.1, and the definitions involved in the statement of the results.

4 Examples of Proof Scripts

Using the techniques in the above two sections, we formalized enough mathematics in Isabelle/FOL to be able to define the fundamental group. In addition to work directly used for that purpose, we also formalized several interesting results on the side. These include the well-ordering theorem and Zorn's lemma, the first isomorphism theorem for groups, and the intermediate value theorem. Two more examples will be presented in the remainder of this section, to demonstrate the level of succinctness of proof scripts that can be achieved.

Throughout our work, we referred to various sources including both mathematical texts and other formalizations. We list these sources here:

- Axioms of set theory and basic operations on sets, construction of natural numbers using least fixed points: from Isabelle/ZF [13, 14].
- Equivalence and order relations, arbitrary products on sets, well-ordering theorem and Zorn's lemma: from Bourbaki's *Theory of Sets* [2].
- Group theory and the construction of real numbers using Cauchy sequences: from my previous case studies [17], which in turn is based on corresponding theories in the Isabelle/HOL library.
- Point-set topology and construction of the fundamental group: from *Topology* by Munkres [12].

4.1 Schroeder-Bernstein Theorem

For our first example, we present the proof of the Schroeder-Bernstein theorem. See [14] for a presentation of the same proof in Isabelle/ZF. The bijection is constructed by gluing together two functions. Auto2 is able to prove automatically that under certain conditions, the gluing is a bijection (lemma `glue_function2_bij`). For the Schroeder-Bernstein theorem, a proof script (provided by the user) is needed. This is given immediately after the statement of the theorem.

```

definition glue_function2 :: "i ⇒ i ⇒ i" where
  "glue_function2(f,g) = Fun(source(f)∪source(g), target(f)∪target(g),
    λx. if x ∈ source(f) then f`x else g`x)"
    
```

```

lemma glue_function2_bij [backward]:
  "f ∈ A ≅ B ⇒ g ∈ C ≅ D ⇒ A ∩ C = ∅ ⇒ B ∩ D = ∅ ⇒
  glue_function2(f,g) ∈ (A ∪ C) ≅ (B ∪ D)"
    
```

theorem schroeder_bernstein:

```
"injective(f) ==> injective(g) ==> f ∈ X → Y ==> g ∈ Y → X ==>
equipotent(X,Y)"
LET "X_A = lfp(X, λW. X - g``(Y - f``W))" THEN
LET "X_B = X - X_A, Y_A = f``X_A, Y_B = Y - Y_A" THEN
HAVE "X - g``Y_B = X_A" THEN
HAVE "g``Y_B = X_B" THEN
LET "f' = func_restrict_image(func_restrict(f,X_A))" THEN
LET "g' = func_restrict_image(func_restrict(g,Y_B))" THEN
HAVE "glue_function2(f', inverse(g')) ∈ (X_A ∪ X_B) ≅ (Y_A ∪ Y_B)"
```

4.2 Rempe-Gillen's Challenge

For our second example, we present our solution to a challenge problem proposed by Lasse Rempe-Gillen in a mailing list discussion². See [1] for proofs of the same result in several other systems. The statement to be proved is:

Lemma 1. *Let f be a continuous real-valued function on the real line, such that $f(x) > x$ for all x . Let x_0 be a real number, and define the sequence x_n recursively by $x_{n+1} := f(x_n)$. Then x_n diverges to infinity.*

Our solution is as follows. We make use of several previously proved results: any bounded increasing sequence in \mathbb{R} converges (line 2), a continuous function f maps a sequence converging to x to a sequence converging to $f \cdot x$ (line 4), and finally that the limit of a sequence in \mathbb{R} is unique.

lemma rempe_gillen_challenge:

```
"real_fun(f) ==> continuous(f) ==> incr_arg_fun(f) ==> x0 ∈. ℝ ==>
S = Seq(ℝ, λn. nfold(f,n,x0)) ==> ¬upper_bounded(S)"
HAVE "seq_incr(S)" WITH HAVE "∀n∈.ℕ. S`n +_ℝ 1 ≥_ℝ S`n" THEN
CHOOSE "x, converges_to(S,x)" THEN
LET "T = Seq(ℝ, λn. f`(S`n))" THEN
HAVE "converges_to(T,f`x)" THEN
HAVE "converges_to(T,x)" WITH (
  HAVE "∀r>_ℝ 0_ℝ. ∃k∈.ℕ. ∀n≥_ℕ k. |T`n -_ℝ x|_ℝ <_ℝ r" WITH (
    CHOOSE "k ∈. ℕ, ∀n≥_ℕ k. |S`n -_ℝ x|_ℝ <_ℝ r" THEN
    HAVE "∀n≥_ℕ k. |T`n -_ℝ x|_ℝ <_ℝ r" WITH HAVE "T`n = S`(n +_ℕ 1)"))
```

5 Construction of the Fundamental Group

In this section, we describe our construction of the fundamental group. We will focus on stating the definitions and main results without proof, to demonstrate the expressiveness of untyped set theory under our framework. The entire formalization including proofs is 864 lines long.

Let I be the interval $[0, 1]$, equipped with the subspace topology from the topology on \mathbb{R} . Given two continuous maps f and g from S to T , a *homotopy* between f and g is a continuous map from the product topology on $S \times I$ to T that restricts to f and g at $S \times \{0\}$ and $S \times \{1\}$, respectively:

² <http://www.cs.nyu.edu/pipermail/fom/2014-October/018243.html>.

definition `is_homotopy` :: "[i, i, i] ⇒ o" where
`"is_homotopy(f,g,F) ←→`
`(let S = source_str(f) in let T = target_str(f) in`
`continuous(f) ∧ continuous(g) ∧`
`S = source_str(g) ∧ T = target_str(g) ∧ F ∈ S ×r I →r T ∧`
`(∀ x ∈ .S. F\r(x, 0ℝ) = f\rx ∧ F\r(x, 1ℝ) = g\rx)"`

A *path* is a continuous function from the interval. A homotopy between two paths is a *path homotopy* if it remains constant on $\{0\} \times I$ and $\{1\} \times I$:

definition `is_path` :: "i ⇒ o" where
`"is_path(f) ←→ (f ∈ I →r target_str(f))"`

definition `is_path_homotopy` :: "[i, i, i] ⇒ o" where
`"is_path_homotopy(f,g,F) ←→`
`(is_path(f) ∧ is_path(g) ∧ is_homotopy(f,g,F) ∧`
`(∀ t ∈ .I. F\r(0ℝ, t) = f\r(0ℝ) ∧ F\r(1ℝ, t) = f\r(1ℝ))"`

Two paths are *path-homotopic* if there exists a path homotopy between them. This is an equivalence relation on paths.

definition `path_homotopic` :: "i ⇒ i ⇒ o" where
`"path_homotopic(f,g) ←→ (∃ F. is_path_homotopy(f,g,F))"`

The path product is defined by gluing two morphisms. It is continuous by the pasting lemma:

definition `I1 = subspace(ℝ, closed_interval(ℝ, 0ℝ, 1ℝ /ℝ 2ℝ))`
definition `I2 = subspace(ℝ, closed_interval(ℝ, 1ℝ /ℝ 2ℝ, 1ℝ))`
definition `interval_lower = Mor(I1, I, λt. 2ℝ *ℝ t)`
definition `interval_upper = Mor(I2, I, λt. 2ℝ *ℝ t -ℝ 1ℝ)`

definition `path_product` :: "i ⇒ i ⇒ i" (infixl "★" 70) where
`"f ★ g = glue_morphism(I, f ◦m interval_lower, g ◦m interval_upper)"`

The loop space is a set of loops on X . Path homotopy gives an equivalence relation on the loop space, and we define `loop_classes` to be the quotient set:

definition `loop_space` :: "i ⇒ i ⇒ i" where
`"loop_space(X,x) = {f ∈ I →r X. f\r(0ℝ) = x ∧ f\r(1ℝ) = x}"`

definition `loop_space_rel` :: "i ⇒ i ⇒ i" where
`"loop_space_rel(X,x) = Equiv(loop_space(X,x), λf g.`
`path_homotopic(f,g))"`

definition `loop_classes` :: "i ⇒ i ⇒ i" where
`"loop_classes(X,x) = loop_space(X,x) // loop_space_rel(X,x)"`

Finally, the fundamental group is defined as:

```

definition fundamental_group :: "i  $\Rightarrow$  i  $\Rightarrow$  i" (" $\pi_1$ ") where
  " $\pi_1(X, x) = (\text{let } \mathcal{R} = \text{loop\_space\_rel}(X, x) \text{ in}$ 
    Group(loop_classes(X, x), equiv_class( $\mathcal{R}$ , const_mor(I, X, x)),
       $\lambda f g. \text{equiv\_class}(\mathcal{R}, \text{rep}(\mathcal{R}, f) \star \text{rep}(\mathcal{R}, g))))$ "

```

To show that the fundamental group is actually a group, we need to show that the path product respects the equivalence relation given by path homotopy, and is associative up to equivalence (along with properties about inverse and identity). The end result is:

```

lemma fundamental_group_is_group:
  " $\text{is\_top\_space}(X) \Longrightarrow x \in. X \Longrightarrow \text{is\_group}(\pi_1(X, x))$ "

```

An important property of the fundamental group is that a continuous function between topological spaces induces a homomorphism between their fundamental groups. This is defined as follows:

```

definition induced_mor :: "i  $\Rightarrow$  i  $\Rightarrow$  i" where
  " $\text{induced\_mor}(k, x) =$ 
    (let X = source_str(k) in let Y = target_str(k) in
      let  $\mathcal{R} = \text{loop\_space\_rel}(X, x)$  in let  $\mathcal{S} = \text{loop\_space\_rel}(Y, k \backslash x)$  in
        Mor( $\pi_1(X, x)$ ,  $\pi_1(Y, k \backslash x)$ ,  $\lambda f. \text{equiv\_class}(\mathcal{S}, k \circ_m \text{rep}(\mathcal{R}, f))$ ))"

```

The induced map is a homomorphism satisfying functorial properties:

```

lemma induced_mor_is_homomorphism:
  " $\text{continuous}(k) \Longrightarrow X = \text{source\_str}(k) \Longrightarrow Y = \text{target\_str}(k) \Longrightarrow$ 
   $x \in \text{source}(k) \Longrightarrow \text{induced\_mor}(k, x) \in \pi_1(X, x) \rightarrow_{\mathcal{G}} \pi_1(Y, k \backslash x)$ "

```

```

lemma induced_mor_id:
  " $\text{is\_top\_space}(X) \Longrightarrow x \in. X \Longrightarrow$ 
   $\text{induced\_mor}(\text{id\_mor}(X), x) = \text{id\_mor}(\pi_1(X, x))$ "

```

```

lemma induced_mor_comp:
  " $\text{continuous}(k) \Longrightarrow \text{continuous}(h) \Longrightarrow$ 
   $\text{target\_str}(k) = \text{source\_str}(h) \Longrightarrow x \in \text{source}(k) \Longrightarrow$ 
   $\text{induced\_mor}(h \circ_m k, x) = \text{induced\_mor}(h, k \backslash x) \circ_m \text{induced\_mor}(k, x)$ "

```

6 Related Work

In Isabelle, the main library for formalized mathematics using FOL is Isabelle/ZF. The basics of Isabelle/ZF is described in [13, 14]. We also point to [13] for a review of older work on set theory from automated deduction and artificial intelligence communities. Outside the official library, IsarMathLib [5] is a more recent project based on Isabelle/ZF. It formalized more results in abstract algebra and point-set topology, and also constructed the real numbers. The initial parts of our development closely parallels that in Isabelle/ZF, but we go further in several directions including constructing the number system. The primary difference between our work and IsarMathLib is that we use auto2

for proofs, and develop our own system for handling structures, so that we do not make use of Isabelle tactics, Isar, or locales.

Outside Isabelle, the major formalization projects using set theory include Metamath [11] and Mizar [4], both of which have extensive mathematical libraries. There are some recent efforts to reproduce the Mizar environment in HOL-type systems [6, 8]. While there are some similarities between our framework and Mizar's, we do not aim for an exact reproduction. In particular, we maintain the typical style of stating definitions and theorems in Isabelle. More comparisons between our approach and Mizar are discussed in Sect. 3.4.

Mizar formalized not just the definition of the fundamental group [7], but several of its properties, including the computation of the fundamental group of the circle. There is also a formalization of path homotopy in HOL Light which is then ported to Isabelle/HOL. This is used for the proof of the Brouwer fixed-point theorem and the Cauchy integral theorem, although the fundamental group itself does not appear to be constructed.

In homotopy type theory, one can work with fundamental groups (and higher-homotopy groups) using synthetic definitions. This has led to formalizations of results about homotopy groups that are well beyond what can be achieved today using standard definitions (see [3] for a more recent example). We emphasize that our definition of the fundamental group, as with Mizar's, follows the standard one in set theory.

7 Conclusion

We applied the auto2 prover to the formalization of mathematics using untyped set theory. Starting from the axioms of set theory, we formalized the definition of the fundamental group, as well as many other results in set theory, group theory, point-set topology, and real analysis. The entire development contains over 13,000 lines of theory files and 3,500 lines of ML code, taking the author about 5 months to complete. On a laptop with two 2.0 GHz cores, it can be compiled in about 24 min. Through this work, we demonstrated the ability of auto2 to scale to relatively large projects. We also hope this result can bring renewed interest to formalizing mathematics in untyped set theory in Isabelle.

Acknowledgements. The author would like to thank the anonymous referees for their comments. This research is completed while the author is supported by NSF Award No. 1400713.

References

1. Blanchette, J.C., Kaliszky, C., Paulson, L.C., Urban, J.: Hammering towards QED. *J. Formalized Reason.* **9**(1), 101–148 (2016)
2. Bourbaki, N.: *Theory of Sets*. Springer, Heidelberg (2000)
3. Brunerie, G.: *On the homotopy groups of spheres in homotopy type theory*. Ph.D. thesis. <https://arxiv.org/abs/1606.05916>

4. Grabowski, A., Kornilowicz, A., Naumowicz, A.: Mizar in a nutshell. *J. Formaliz. Reason. Spec. Issue: User Tutor. I* **3**(2), 153–245 (2010)
5. IsarMathLib. <http://www.nongnu.org/isarmathlib/>
6. Kaliszzyk, C., Pak, K., Urban, J.: Towards a Mizar environment for Isabelle: foundations and language. In: *Proceedings of the 5th ACM SIGPLAN Conference on Certified Programs and Proofs (CPP 2016)*, New York, pp. 58–65 (2016)
7. Kornilowicz, A., Shidama, Y., Grabowski, A.: The fundamental group. *Formalized Math.* **12**(3), 261–268 (2004)
8. Kuncar, O.: Reconstruction of the Mizar type system in the HOL light system. In: Pavlu, J., Safrankova, J. (eds.) *WDS Proceedings of Contributed Papers: Part I - Mathematics and Computer Sciences*, pp. 7–12. Matfyzpress (2010)
9. Lee, G., Rudnici, P.: Alternative aggregates in Mizar. In: Kauer, M., Kerber, M., Miner, R., Windsteiger, W. (eds.) *Calculemus/MKM 2007. LNCS (LNAI)*, vol. 4573, pp. 327–341. Springer, Heidelberg (2007). doi:[10.1007/978-3-540-73086-6_26](https://doi.org/10.1007/978-3-540-73086-6_26)
10. Mahboubi, A., Tassi, E.: Canonical structures for the working Coq user. In: Blazy, S., Paulin-Mohring, C., Pichardie, D. (eds.) *ITP 2013. LNCS*, vol. 7998, pp. 19–34. Springer, Heidelberg (2013). doi:[10.1007/978-3-642-39634-2_5](https://doi.org/10.1007/978-3-642-39634-2_5)
11. Megill, N.D.: *Metamath: a computer language for pure mathematics*. <http://us.metamath.org/downloads/metamath.pdf>
12. Munkres, J.R.: *Topology*. Prentice Hall, Upper Saddle River (2000)
13. Paulson, L.C.: Set theory for verification: I. From foundations to functions. *J. Automated Reason.* **11**(3), 353–389 (1993)
14. Paulson, L.C.: Set theory for verification: II. Induction and recursion. *J. Automated Reason.* **15**(2), 167–215 (1995)
15. Trybulec, A.: Some features of the Mizar language. In: *ESPRIT Workshop* (1993)
16. Wiedijk, F.: Mizar’s soft type system. In: Schneider, K., Brandt, J. (eds.) *TPHOLS 2007. LNCS*, vol. 4732, pp. 383–399. Springer, Heidelberg (2007). doi:[10.1007/978-3-540-74591-4_28](https://doi.org/10.1007/978-3-540-74591-4_28)
17. Zhan, B.: AUTO2, a saturation-based heuristic prover for higher-order logic. In: Blanchette, J.C., Merz, S. (eds.) *ITP 2016. LNCS*, vol. 9807, pp. 441–456. Springer, Cham (2016). doi:[10.1007/978-3-319-43144-4_27](https://doi.org/10.1007/978-3-319-43144-4_27)